NAG Library Routine Document

G01EFF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

G01EFF returns the lower or upper tail probability of the gamma distribution, with parameters α and β, via the routine name.

2 Specification

```
FUNCTION G01EFF (TAIL, G, A, B, IFAIL)
REAL (KIND=nag_wp) G01EFF
INTEGER            IFAIL
REAL (KIND=nag_wp) G, A, B
CHARACTER(1)       TAIL
```

3 Description

The lower tail probability for the gamma distribution with parameters α and β, $P(G \leq g)$, is defined by:

$$P(G \leq g;\alpha,\beta) = \frac{1}{\beta^{\alpha}\Gamma(\alpha)}\int_0^g G^{\alpha-1}e^{-G/\beta}\,dG, \quad \alpha > 0.0,\ \beta > 0.0.$$

The mean of the distribution is $\alpha\beta$ and its variance is $\alpha\beta^2$. The transformation $Z = \dfrac{G}{\beta}$ is applied to yield the following incomplete gamma function in normalized form,

$$P(G \leq g;\alpha,\beta) = P(Z \leq g/\beta : \alpha, 1.0) = \frac{1}{\Gamma(\alpha)}\int_0^{g/\beta} Z^{\alpha-1}e^{-Z}\,dZ.$$

This is then evaluated using S14BAF.

4 References

Hastings N A J and Peacock J B (1975) *Statistical Distributions* Butterworth

5 Arguments

1: TAIL – CHARACTER(1) *Input*

On entry: indicates whether an upper or lower tail probability is required.

TAIL = 'L'
The lower tail probability is returned, that is $P(G \leq g : \alpha, \beta)$.

TAIL = 'U'
The upper tail probability is returned, that is $P(G \geq g : \alpha, \beta)$.

Constraint: TAIL = 'L' or 'U'.

2: G – REAL (KIND=nag_wp) *Input*

On entry: g, the value of the gamma variate.

Constraint: G $\geq$ 0.0.

3: A – REAL (KIND=nag_wp) *Input*

On entry: the parameter α of the gamma distribution.

Constraint: A > 0.0.

4: B – REAL (KIND=nag_wp) *Input*

On entry: the parameter β of the gamma distribution.

Constraint: B > 0.0.

5: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this argument you should refer to Section 3.4 in How to Use the NAG Library and its Documentation for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this argument, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: IFAIL $= 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

If IFAIL $= 1$, 2, 3 or 4 on exit, then G01EFF returns 0.0.

IFAIL $= 1$

On entry, TAIL $\neq$ 'L' or 'U'.

IFAIL $= 2$

On entry, G < 0.0.

IFAIL $= 3$

On entry, A ≤ 0.0,
or B ≤ 0.0.

IFAIL $= 4$

The solution did not converge in 600 iterations. See S14BAF. The probability returned should be a reasonable approximation to the solution.

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.9 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.8 in How to Use the NAG Library and its Documentation for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.7 in How to Use the NAG Library and its Documentation for further information.

7 Accuracy

The result should have a relative accuracy of ***machine precision***. There are rare occasions when the relative accuracy attained is somewhat less than ***machine precision*** but the error should not exceed more than 1 or 2 decimal places. Note also that there is a limit of 18 decimal places on the achievable accuracy, because constants in S14BAF are given to this precision.

8 Parallelism and Performance

G01EFF is not threaded in any implementation.

9 Further Comments

The time taken by G01EFF varies slightly with the input arguments G, A and B.

10 Example

This example reads in values from a number of gamma distributions and computes the associated lower tail probabilities.

10.1 Program Text

```
    Program g01effe

!     G01EFF Example Program Text

!     Mark 26 Release. NAG Copyright 2016.

!     .. Use Statements ..
      Use nag_library, Only: g01eff, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Real (Kind=nag_wp)               :: a, b, g, p
      Integer                          :: ifail
      Character (1)                    :: tail
!     .. Executable Statements ..
      Write (nout,*) 'G01EFF Example Program Results'
      Write (nout,*)

!     Skip heading in data file
      Read (nin,*)

!     Display titles
      Write (nout,*) ' TAIL        G            A        B      Probability'
      Write (nout,*)

d_lp: Do
        Read (nin,*,Iostat=ifail) tail, g, a, b
        If (ifail/=0) Then
          Exit d_lp
        End If

!       Calculate probability
        ifail = 0
        p = g01eff(tail,g,a,b,ifail)
```

```
!       Display results
        Write (nout,99999) tail, g, a, b, p
      End Do d_lp

99999 Format (3X,A1,4X,F9.2,F13.2,F9.2,7X,F7.4)
    End Program g01effe
```

10.2 Program Data

```
G01EFF Example Program Data
'L' 15.5   4.0   2.0 :TAIL G A B
'L'  0.5   4.0   1.0
'L' 10.0   1.0   2.0
'L'  5.0   2.0   2.0
```

10.3 Program Results

```
 G01EFF Example Program Results

  TAIL        G            A        B      Probability

   L        15.50         4.00     2.00        0.9499
   L         0.50         4.00     1.00        0.0018
   L        10.00         1.00     2.00        0.9933
   L         5.00         2.00     2.00        0.7127
```